Einsatz von FPW-Sourcen unter VFP 
"Eins für 2 und 3"
Manfred Rätzmann 

Manfred Rätzmann beschreibt die Einsatz- und Erweiterungsmöglichkeiten von FoxPro/Windows-2.6-Sourcecode unter Visual FoxPro 3.0 und höheren Versionen, basierend auf Ideen von Alan Schwartz, dem bekannten Autor des Berichtswerkzeuges FoxFire!. 



Jeder, der eine größere Anwendung unter FP 2.x entwickelt hat und jetzt mit Visual FoxPro rumprobiert, wird das Dilemma kennen: Am liebsten würde man noch einmal komplett von vorne anfangen - aber es geht nicht. Wer hat die Zeit dazu, wer soll das bezahlen? Anderseits ist VFP doch zu FP2.x abwärtskompatibel. Also sollte es möglich sein, alle Programme der 2.x-Anwendung inclusive der von GENSCRN generierten .SPR in ein VFP-Projekt zu packen, dort neu zu kompilieren und dann ablaufen zu lassen. Gesagt, getan und ausprobiert ... läuft - na ja, fast! Etwas enttäuscht wollte ich diese Idee schon zu den Akten legen, als mir ein Artikel von Alan C. Schwartz im FoxPro Advisor auffiel. Alan beschreibt dort, wie sie genau diesen Weg gegangen sind, um FoxFire! 3.0 sowohl unter FP 2.6 als auch unter VFP 3.0 zur Verfügung zu stellen. Mehr noch: die FoxFire!-Crew hat einen Weg gefunden, Masken, die unter 2.6 erstellt wurden, in 3.0 als Objekte anzusprechen und zu manipulieren. Am Thema der Listboxen zeigt er, wie ein Programm, das unter 2.6 entwickelt wird, trotzdem diverse VFP-Neuerungen nutzen kann - zum Beispiel die mehrspaltigen Listboxen. Ich habe die dort beschriebene Vorgehensweise an meinem WWFORM-Programm ausprobiert und bin davon sehr angetan. 

WINDOW-Definitionen mit Anfasser 

Die zentrale Idee ist, die unter VFP verfügbare NAME-Klausel für DEFINE WINDOW und das @ GET-Statement zu nutzen. Mit DEFINE WINDOW ... NAME Fenster kann man einem definierten Fenster einen Objektnamen mitgeben, unter dem das Fenster jetzt ansprechbar ist. Geben Sie einfach mal folgendes im Befehlsfenster ein: 

define window dummy from 1,10 to 20,50; 
system float grow zoom close name Fenster 
activate window dummy 

"Fenster" ist jetzt Ihre Objektreferenz (oder Handle, zu deutsch: Anfasser) für das angezeigte Window. Also können Sie nun mit 

Fenster.Caption="Mein Fenstertitel" 

den Titel des Fensters setzen oder mit 

Fenster.Backcolor=RGB(0,150,0) 

dem Fenster ein sattes Grün als Hintergrund verpassen. Nicht nur das Setzen der Eigenschaften funktioniert so, auch der Aufruf von Methoden: 

Fenster.Move(1,1,200,200) 

verschiebt das Fenster und gibt ihm eine andere Größe. So wird mit Hilfe der NAME-Klausel aus einem 2.6er Window ein 3.0-Objekt, das mit normalen VFP-Befehlen manipuliert werden kann. Diese Befehle sollten dann mit Hilfe der Präprozessoranweisungen #IF - #ENDIF im 2.6er Programm ausgeblendet werden. Schon hätte man die Möglichkeit, die gleiche Source unter 2.6 und - leicht ergänzt - unter 3.0 zu kompilieren und ausführen zu lassen. Wenn man nur dem GENSCRN beibringen könnte, diese NAME-Klausel in das DEFINE WINDOW mit einzubauen - allerdings nur auf der VFP-Seite! Aber ich greife vor, meine Begeisterung trägt mich davon. Wie die Jungs von FoxFire! das geschafft haben, ist tatsächlich Spitze! 

Eingelegte @GET-Felder 

Schauen wir uns vorher noch das Verhalten von @ GET Feldern mit NAME-Klausel an. Dazu verpassen wir unserem FENSTER zunächst mal ein solches @ GET-Feld. Um alle im Moment existierenden Objekte in Ruhe untersuchen zu können, habe ich mir ein kleines Programm gebastelt, das ich DISPOBJE.PRG genannt habe: 

private all like p* 
pcSafety = set("SAFETY") 
set safety off 
disp obje to file objekte.txt noco 
modi file objekte.txt 
set safety &pcSafety 

Mit Hilfe dieses Programms schaue ich mir FENSTER zunächst einmal an. Es zeigt mir, daß FENSTER ein ganz normales Objekt der Klasse FORM ist, ohne alle Unterobjekte. Nachdem ich den Befehl 

@ 2,2 get m.Eingabe name Eingabe 

im Befehlsfenster eingegeben habe, zeigt mir DISPOBJE, daß die Form FENSTER nunmehr ein Unterobjekt PAGEFRAME1 vom Typ PAGEFRAME enthält. Die PAGE1 dieses PAGEFRAME1 beinhaltet nun mein @ GET-Feld. Über 

Fenster.pageframe1.page1.Eingabe.Value="Hallo" 

kann ich diesem @ GET-Feld einen Wert zuweisen. (DISPOBJE zeigt mir allerdings kein Objekt mit Namen EINGABE. Auch die PAGE1 aus PAGEFRAME1 wird nicht angegeben. Offensichtlich taucht der DISPLAY OBJECTS Befehl nicht in die Tiefen einer Containerstruktur.) 

Um auch das Objekt zu meinem Eingabefeld näher betrachten zu können (und um mir das Eintippen der langen Objektreferenz zu ersparen), besorge ich mir mit 

Feld=Fenster.Pageframe1.page1.Eingabe 

einen Verweis auf das gerade angelegte Eingabefeld. Aha: FELD ist ein TEXTBOX-Objekt mit dem aktuellen VALUE = "Hallo". Aber in dem Feld in der Maske wird nichts angezeigt. Auch ein Aufruf der Methoden REFRESH(), SETFOCUS() oder CLICK() funktioniert so nicht. Wenn man allerdings ein READ im Befehlsfenster eingibt, erscheint "Hallo" im Eingabefeld, das jetzt auch den Focus hat und eine Eingabe ermöglicht. Wenn der READ beendet wird, sind PAGEFRAME1 und mein Eingabefeld wieder verschwunden. Visual FoxPro führt im Kompatibilitätsmodus anscheinend nicht alle Methoden der so erstellten Objekte aus. Alan C. Schwartz führt in seinem Artikel noch einige andere Stolpersteine auf, die aber mit wenig Aufwand aus dem Weg geräumt werden können. Der Gewinn bei der Sache ist die Möglichkeit, fertige und funktionierende 2.x-Programme, die sich nicht von heute auf morgen neu schreiben lassen, auch unter VFP 3.0 einsetzen zu können. 

Wir schieben dem Fuchs ein Kuckucksei unter 

Wie schafft man es jetzt, daß der vom GENSCRN unter FP 2.x generierte Code die für die Handhabung in VFP 3.0 so sinnvolle NAME-Klausel generiert? Zur Veranschaulichung hier der Code, der in der INIT-Prozedur einer WWFORM-Maske abgelegt ist: 

#section 1 
if vers() = "Visual" 
frmNameCls = " NAME mskForm " 
getName001 = " NAME lstForms " 
else 
frmNameCls = " " 
getName001 = " " 
endif 
#itsexpression ~ 
wtitle = "Formulare erstellen und bearbeiten" 

#section 2 
#if vers() = "Visual" 
mskForm.lockscreen = .T. 
#endif 

Schauen wir uns die einzelnen Zeilen an. #section1 besagt, daß der folgende Code in der SPR-Datei vor dem DEFINE WINDOW Kommando erscheinen soll. Mit if vers() = "Visual" wird abgeprüft, ob das kompilierte Programm gerade unter Visual FoxPro oder unter einer Vorversion abläuft. Beim Ablauf unter VFP werden in diesem Fall zwei Variablen mit Werten gefüllt, die unter FP 2.x leer bleiben sollen. Diese Variablen beinhalten den NAME-Zusatz. Wie wird dieser Zusatz jetzt in die Definition des WINDOWS und des @ GET-Feldes geschmuggelt? 

Der DEFINE WINDOW Befehl wird über die TITLE-Klausel infiltriert. #itsexpression ~ teilt GENSCRN mit, daß ein mit der Tilde beginnender Fenstertitel kein Literal, sondern ein Ausdruck ist. Im Feld "Titel" der Maskenlayout-Maske (Menüpunkt Maske, Unterpunkt Layout im FP 2.6 Maskenentwurfsfenster) wird ~wtitle &frmNameCls als Maskentitel angegeben. In der SPR-Datei wird daraus: 

DEFINE WINDOW _r9w0yeujb ; 
AT 0.000, 0.000 ; 
SIZE 23.143,136.200 ; 
TITLE wtitle &frmNameCls ; 
FONT "Arial", 8 ; 
FLOAT ; 
NOCLOSE ; 
MINIMIZE ; 
SYSTEM ; 
COLOR RGB(,,,192,192,192) 

Somit haben wir in die TITLE-Zeile des DEFINE WINDOW ein Makro eingeschmuggelt. Da dieses Makro zur Laufzeit unter VFP mit " NAME mskForm " vorbesetzt ist, lautet der dort ausgeführte Befehl 

DEFINE WINDOW _r9w0yeujb ; 
... 
TITLE wtitle NAME mskForm ; 
... 

Unter dem Namen mskForm kann sich also jeglicher VFP 3.0 spezifische Code auf diese Maske als ein Objekt beziehen. 

Bei @ GET-Feldern wird die Message-Klausel benutzt, um dem Screen-Objekt einen Objektnamen anzuhängen. Dazu muß die Message als Ausdruck gekennzeichnet werden. Der Ausdruck, der in unserem Beispiel der WWFORM-Maske in der Message-Klausel des Listenfeldes hinterlegt ist, lautet 

" " &getName001 

GENSCRN macht daraus in der SPR-Datei folgendes: 

@ 1.786,0.800 GET jcAnzZeile ; 
... 
MESSAGE " " &getName001 ; 
... 

Auch hier erfolgt zur Laufzeit also eine Makrosubstitution, die unter FP 2.x ein leeres Makro in die Zeile einträgt, unter VFP 3.0 die Zeile jedoch zu MESSAGE " " NAME lstForms abändert. Das Listenfeld kann somit unter VFP 3.0 als Objekt mit dem Namen lstForms angesprochen werden. 

Und was haben wir nun davon? 

Jetzt haben wir eine mit dem FP 2.x Maskengenerator erstellte und bearbeitbare Maske zur Verfügung, die nach wie vor einwandfrei funktioniert. Die generierte SPR-Datei kann in ein VFP-Projekt eingebunden werden. Die dabei über Macrosubstitution eingeflochtenen NAME-Klauseln ermöglichen es uns, auf die Maske oder einzelne Bestandteile daraus als (fast) normale VFP-Objekte zuzugreifen. Im SHOW GETS - Snippet habe ich zum Beispiel folgenden Code untergebracht: 

#IF vers() = "Visual" 
_screen.activeform.lockscreen = .T. 
private jcSeite 
jcSeite = "Page"+ ; 
alltrim(str(mskForm.PageFrame1.ActivePage,1)) 
with mskForm.pageframe1.&jcSeite..lstForms 
if .columncount <> 2 
.columncount = 2 
.columnwidths = "560,80" 
.fontname="Arial" 
.fontsize=9 
.columnlines=0 
.rowsource="wwf1.wwf1_bez,wwf1_datei" 
.rowsourcetype=6 
endif 
endwith 
_screen.activeform.lockscreen = .F. 
#ENDIF 

Zunächst stelle ich sicher, daß der folgende Code nur unter VFP kompiliert wird. Dazu benutze ich die Präprozessoranweisung #IF vers() = "Visual". Das funktioniert, weil die so ausgeklammerten Anweisungen komplett in die SPR-Datei übernommen werden. Erst beim Kompilieren der SPR-Datei wird die Präprozessoranweisung #IF wirksam. Also genau so, wie wir das brauchen. Das habe ich mir übrigens auch im INIT-Snippet zu nutze gemacht. Damit die Maske gar nicht erst in der nicht-bearbeiteten Form angezeigt wird, setze ich unter VFP lockscreen zunächst auf .T. Beim ersten Anzeigen der Maske wird auch die SHOW GETS-Routine durchlaufen. Dort wird nach dem Geraderücken der Listenfeld-Properties lockscreen auf .F. gesetzt. 

Zwischen WITH und ENDWITH setze ich die Properties des Listenfeldes so, wie sie unter VFP erscheinen sollen. Da das nur beim ersten Durchlauf passieren soll, prüfe ich ab, ob columncount bereits auf 2 steht. Wenn nicht, wird aus der einspaltigen FP 2.x Listbox eine zweispaltige VFP Listbox, die unter anderem die Möglichkeit hat, einen Proportionalfont wie ARIAL sauber anzuzeigen. Die Breite der benötigten Spalten ermittelt man am besten, indem man den Programmablauf in VFP unterbricht und aus dem Befehlsfenster heraus verschiedene Spaltenbreiten des nun äußerst zugänglichen Listbox-Objekts durchprobiert. 

Bugs und Workarounds 

Die seltsame Konstruktion zur Ermittlung der Seitennummer aus dem ActivePage Property dient dazu, ein Verhalten von VFP abzufangen, hinter dem Alan C. Schwartz wahrscheinlich zu Recht einen Bug vermutet: Wenn eine Maske im Kompatibilitätsmodus aus einer anderen Maske heraus aufgerufen wird, also ein "nested READ" vorliegt, so ist die Seite im Pageframe, die die Screen-Objekte enthält, plötzlich nicht mehr die PAGE1 sondern PAGE2. Indem ich die Seitennummer aus der ActivePage ermittle, brauche ich nicht PAGE1 oder PAGE2 direkt im Sourcecode anzugeben. 

Besonderes Augenmerk sollten Sie auch auf BROWSE-Fenster legen, die aus einer Maske heraus aufgerufen werden. Beim Schließen eines solchen BROWSE-Fensters kann es passieren, daß die WHEN-Klausel eines Listenfelds der aufrufenden Maske durchlaufen wird. Bei meinen Versuchen passierte das immer dann, wenn das Listenfeld das erste aktive Objekt der aufrufenden Maske war oder wenn ich das BROWSE-Fenster durch Anklicken der darunterliegenden Maske geschlossen und dabei zufällig auf das Listenfeld geklickt habe. Dieses Verhalten habe ich umschifft, indem ich einen Merker benutze, der nur dann auf .T. gesetzt wird, wenn das Listenfeld angewählt werden darf. 

Ein Button, der mit \! als Standardbutton gekennzeichnet ist, reagiert unter VFP auch, wenn Sie mit ENTER eine Eingabe in irgendein anderes Feld abschließen. Unter FP 2.6 war das nicht immer so, wie ich jetzt gemerkt habe. Deshalb benutze ich diese Setzung nicht mehr. 

Zum Schluß noch eine gute Nachricht. Auch SPR-Dateien, die mit GENSCRNX verändert wurden, können auf diese Weise in ein VFP-Projekt eingebunden werden. Wenn Sie zum Beispiel Ihre Masken unter FPW 2.6 auf 3D gestylt haben, binden Sie 3DBOX.PRG einfach mit in das VFP-Projekt ein. Wie das Ganze bei Benutzung des TABS-Drivers und anderen GENSCRNX-Anwendungen aussieht, weiß ich nicht. Probieren Sie's aus!

Manfred Rätzmann entwickelt seit vielen Jahren mit Visual FoxPro. Er ist regelmäßiger Redner auf der VFP Entwicklerkonferenz in Frankfurt/Main und veröffentlicht Artikel rund um Visual FoxPro. Seine Schwerpunkte sind die Datenmodellierung und die Entwicklung von Komponenten; sein Hauptprodukt ist die Komponentensammlung „THE LIB". Sie erreichen ihn unter 100322.2534@compuserve.com oder im dFPUG-Forum auf CompuServe.

